Diode SPICE Model

- DC equivalent circuit used for SPICE model of pn diode:



- i-v characteristics modeled by 2 parameters: the saturation current (IS) and the ideality factor (N)

$$i_D = IS\left(e^{v_D'/NV_{th}} - 1\right)$$

Diode SPICE Model

- Ideality factor accounts for generation-recombination in the depletion region of the diode (ignored in our derivation for i_D)

- Usual range for N is 0.5 - 1

- R_S is the series resistance of the quasineutral p and n regions plus the ohmic contacts

- True junction voltage given by: $v_D' = v_D - i_D R_S$

- Charge storage in the diode is modeled by the depletion capacitance C_j

Diode SPICE Model

- Complete list of SPICE diode model parameters:

Parameter	Symbol	SPICE name	Units	Default
Saturation current	I_0 or I_S	IS	A	10^{-14}
Emission coefficent	n or N	N	-	1
Series resistance	R_S	RS	Ω	0
Built-in voltage	V_{bi} or ϕ_j	VJ	V	1
Junction Capacitance	C_{j0}	CJ0	F	0
Grading coefficient	m	M	-	0.5
Transit time	τ_t	TT	s	0
Breakdown voltage	V_{BD}	BV	V	∞
Reverse current at breakdown	I_{BD}	IBV	A	10^{-10}

Diode SPICE Model

- Diode included in circuit using the following line:

DXXX NP NN DNAME

where: XXX = diode number
NP, NN = positive and negative diode nodes
DNAME = model name

- Diode model requires a .MODEL statement:

.MODEL DNAME TYPE (PNAME1=PVAL1 PNAME2=PVAL2 ...)

where: TYPE D
PNAME1 = is model parameter 1, with value PVAL1, etc.

Example:

.MODEL DMOD D (IS=1E-17 RS=20 CJO=1P TT = 50N)

Brief Introduction to SPICE

- SPICE = “Simulation Program with Integrated Circuit Emphasis”
- A general purpose program that simulates electronic circuits.
- *PSpice* = PC version of SPICE (from OrCAD Corporation)

Brief Introduction to SPICE

- Types of analysis supported:
  - DC sweeps of current/voltage sources (.DC)
  - operating point determination (.OP)
  - Thevenin equivalents (.TF)
  - time-domain (transient) response (.TRAN)
  - Fourier analysis (.FOUR)
  - small-signal frequency response (.AC)
  - noise analysis (.NOISE)
  - sensitivity analysis (.SENS)
  - etc.

General Procedure for Using SPICE

- 3 basic steps:

  1) create a source file for the circuit to be simulated

  2) enter the source file into the computer to run the program

  3) instruct the computer to print our plot the results

- Educational version of PSpice is available for downloading free from the web site:

  http://www.orcad.com/Product/Simulation/PSpice/eval.asp

General Procedure for Using SPICE

- For specific syntax of input file statements, consult the appropriate reference

- A few good references are:

[1] P. Tuinenga, *A Guide to Circuit Simulation and Analysis Using PSpice*, 2nd Ed., Englewood, NJ: Prentice Hall, 1992.

[2] S. Reidel and J. Nilsson, *Introduction to PSpice*, Menlo Park, CA: Addison Wesley, 1997.

[3] M. Rashid, *SPICE for Circuits and Electronics Using PSpice*, Englewood, NJ: Prentice Hall, 1995.

SPICE Example 1

- Consider the DC circuit below:



- Suppose we want to calculate and print all node voltages and the current and power of all voltage sources (V_S, V_X, and V_Y).

SPICE Example 1

Input File

- Must have five key parts:

    1) Title

    2) Circuit Description

    3) Analysis Requests

    4) Output Requests

    5) End of Program (.END)

SPICE Example 1

Sample Input File:

```
** DC Circuit Example 1 **

** Circuit Description **
VS  1  0  DC  20V    ; DC voltage source of 20 volts
IS  0  4  DC  50MA ; DC cuurent source of 50 milliamps
R1  1  2  500          ; Resistance of 500 ohms
R2  2  5  800          ; Resistance of 800 ohms
R3  2  3  1KOHM     ; Resistance of 1 kilo-ohm
R4  4  0  200          ; Resistance of 200 ohms

** Analysis Requests **
VX  3  0  DC  0V      ; Measures current through R3
VY  5  4  DC  0V      ; Measures current through R2

** Output Requests **
.OP                          ; Directs the bias point to the output file
.END                        ; End of circuit file
```

SPICE Example 1

Running the program:

- ❑ Goto: Start -> Programs -> PSpice Student-> PSpice AD Student
- ❑ From “File” menu, goto: New -> Test File
- ❑ Type in your input file
- ❑ Save your input file with a .CIR extension
- ❑ From “Simulation” menu, goto Run <*input_file*>

❍ The output file will autmatically be given the extension:
<*input_file*>.OUT

❍ The simulation results (i.e., plots) file will have the extension:
<*input_file*>.DAT

SPICE Example 1

Viewing program output:

- To view the output file:

  From "View" menu, goto "Output File"

- To view the simulation results (i.e., plots):
  - From "View" menu, goto "Simulation Results"
  - From "Trace" menu, goto "Add Trace"
  - Select the desired voltage and/or current node

SPICE Example 1

Results appearing in <input_file>.OUT:

```
****     SMALL-SIGNAL BIAS SOLUTION        TEMPERATURE =   27.000 DEG C
 NODE   VOLTAGE     NODE   VOLTAGE     NODE   VOLTAGE     NODE   VOLTAGE
(    1)   20.0000  (    2)   12.5000  (    3)    0.0000  (    4)   10.5000
(    5)   10.5000
    VOLTAGE SOURCE CURRENTS
    NAME          CURRENT
    VS           -1.500E-02                        IR1 = 15 mA

    VX            1.250E-02                        IR3 = 12.5 mA

    VY            2.500E-03                        IR2 = 2.5 mA
    TOTAL POWER DISSIPATION   3.00E-01  WATTS
          JOB CONCLUDED
          TOTAL JOB TIME            1.04
```

SPICE Example 2

- For the same circuit, suppose we want to calculate and print the voltage at node 4, and the currents through R_2 and R_3 for V_S = 10 V, 20 V, and 30 V.

- <u>*Solution*</u>:

1. The input file is the same as above, except that the .OP statement is replaced by:

.DC VS 10V 30V 10V ; Sweep VS from 10 V to 30V in 10V increments

and after that, we add one additional statement:

.PRINT DC V(4) I(VX) I(VY) ; Prints the results of the DC sweep

SPICE Example 2

New Input File:

```
** DC Circuit Example 2 **

** Circuit Description **
VS  1  0  DC  20V    ; DC voltage source of 20 volts
IS  0  4  DC  50MA ; DC cuurent source of 50 milliamps
R1  1  2  500           ; Resistance of 500 ohms
R2  2  5  800           ; Resistance of 800 ohms
R3  2  3  1KOHM     ; Resistance of 1 kilo-ohm
R4  4  0  200           ; Resistance of 200 ohms

** Analysis Requests **
VX  3  0  DC  0V      ; Measures current through R3
VY  5  4  DC  0V      ; Measures current through R2

** Output Requests **
.DC  VS  10V  30V  10V
.PRINT  DC  V(4)  I(VX)  I(VY)
.END                        ; End of circuit file
```

SPICE Example 2

Solution (cont).

2. The program is run in the same way as above (except that we might want to use a different name for our new *input_file*).

3. The resulting <input_file>.out now looks like:

```
****     DC TRANSFER CURVES                    TEMPERATURE =  27.000 DEG C

 VS            V(4)          I(VX)         I(VY)

  1.000E+01     9.500E+00     7.500E-03    -2.500E-03

  2.000E+01     1.050E+01     1.250E-02     2.500E-03

  3.000E+01     1.150E+01     1.750E-02     7.500E-03

          JOB CONCLUDED

          TOTAL JOB TIME             1.21
```

SPICE Example 3

- For the same circuit, suppose we want to calculate and *plot* the current through R_2 for V_S = 10 V, 30 V, and 50 V.

- *Solution*:

1. The input file is the same as Example 1, except that the .OP statement is replaced by:

.DC VS 10V 50V 20V ; Sweep VS from 10 V to 50V in 20V increments

.PLOT DC I(VY) ; Plots the results of the DC sweep
.PROBE ; Invokes PSPICE graphical waveform analyzer

SPICE Example 3

New Input File:

```
** DC Circuit Example 3 **

** Circuit Description **
VS  1  0  DC  20V    ; DC voltage source of 20 volts
IS  0  4  DC  50MA ; DC cuurent source of 50 milliamps
R1  1  2  500           ; Resistance of 500 ohms
R2  2  5  800           ; Resistance of 800 ohms
R3  2  3  1KOHM     ; Resistance of 1 kilo-ohm
R4  4  0  200           ; Resistance of 200 ohms

** Analysis Requests **
VX  3  0  DC  0V      ; Measures current through R3
VY  5  4  DC  0V      ; Measures current through R2

** Output Requests **
.DC  VS  10V  50V  20V
.PLOT DC I(VY)
.PROBE
.END                          ; End of circuit file
```

SPICE Example 3

2. The program is run in the same way as above, except that we now want to view the simulation results using the "Trace" menu.

3. The desired plot looks like: